Module 20.2: nag_t_dist

Probabilities and Deviate for a Student's t-distribution

nag_t_dist provides procedures for computing probabilities and the deviate for various parts of a Student's t-distribution.

Contents

Procedure: nag_t_prob

1 Description

nag_t_prob calculates the lower tail, upper tail or two tail probabilities for a Student's t-distribution with ν degrees of freedom.

2 Usage

```
USE nag_t_dist
[value =] nag_t_prob(tail, t, df  [, optional arguments])
```

The function result is a scalar of type real(kind=*wp*).

3 Arguments

3.1 Mandatory Arguments

tail — character(len=1), intent(in)

Input: the type of tail probability to be returned:

if tail = 'L' or 'l', the lower tail probability is returned;

if tail = 'U' or 'u', the upper tail probability is returned;

if tail = 'S' or 's', the two tail (significance level) probability is returned;

if tail = 'C' or 'c', the two tail (confidence interval) probability is returned.

Constraints: tail = 'L', 'l', 'U', 'u', 'S', 's', 'C' or 'c'.

t — real(kind=*wp*), intent(in)

Input: the value of the Student's t-variate.

df — real(kind=*wp*), intent(in)

Input: the degrees of freedom, ν, of the Student's t-distribution.

Constraints: df ≥ 1.0.

3.2 Optional Argument

error — type(nag_error), intent(inout), optional

The NAG *fl*90 error-handling argument. See the Essential Introduction, or the module document nag_error_handling (1.2). You are recommended to omit this argument if you are unsure how to use it. If this argument is supplied, it *must* be initialized by a call to nag_set_error before this procedure is called.

4 Error Codes

Fatal errors (error%level = 3):

error%code	Description
301	An input argument has an invalid value.

5 Examples of Usage

A complete example of the use of this procedure appears in Example 1 of this module document.

6 Further Comments

6.1 Mathematical Background

Given that variate t follows a Student's t-distribution with ν degrees of freedom, the lower tail probability $P(T \leq t : \nu)$, for instance, is defined by

$$P(T \leq t : \nu) = \frac{\Gamma((\nu+1)/2)}{\sqrt{\pi\nu}\Gamma(\nu/2)} \int_{-\infty}^{t} \left(1 + \frac{T^2}{\nu}\right)^{-(\nu+1)/2} dT, \quad \nu \geq 1.$$

6.2 Algorithmic Detail

Computationally, there are two situations.

(a) When $\nu < 20$, a transformation of the beta distribution, $P_\beta(B \leq \beta : a, b)$, is used:

$$P(T \leq t : \nu) = \frac{1}{2} P_\beta \left(B \leq \frac{\nu}{\nu + t^2} : \frac{\nu}{2}, \frac{1}{2}\right) \text{ when } t < 0.0$$

or

$$P(T \leq t : \nu) = \frac{1}{2} + \frac{1}{2} P_\beta \left(B \geq \frac{\nu}{\nu + t^2} : \frac{\nu}{2}, \frac{1}{2}\right) \text{ when } t > 0.0.$$

(b) When $\nu \geq 20$, an asymptotic normalising expansion of the Cornish–Fisher type is used to evaluate the probability, see Hill [1].

6.3 Accuracy

The result is relatively accurate to 5 significant figures especially for reasonable probabilities but some loss of accuracy is inevitable for very low probabilities (say, less than 10^{-10}).

Procedure: nag_t_deviate

1 Description

nag_t_deviate returns the deviate associated with the lower tail, upper tail, or two tail probability of a Student's t-distribution with ν degrees of freedom.

2 Usage

```
USE nag_t_dist
[value =] nag_t_deviate(tail, p, df  [, optional arguments])
```

The function result is a scalar of type real(kind=*wp*).

3 Arguments

3.1 Mandatory Arguments

tail — character(len=1), intent(in)

Input: indicates which tail the supplied probability represents:

if tail = 'L' or 'l', p contains the lower tail probability;

if tail = 'U' or 'u', p contains the upper tail probability;

if tail = 'S' or 's', p contains the two tail (significance level) probability;

if tail = 'C' or 'c', p contains the two tail (confidence interval) probability.

Constraints: tail = 'L', 'l', 'U' or 'u', 'S', 's', 'C' or 'c'.

p — real(kind=*wp*), intent(in)

Input: the probability (as defined by tail) for the required Student's t-distribution.

Constraints: $0.0 < \mathtt{p} < 1.0$.

df — real(kind=*wp*), intent(in)

Input: the degrees of freedom, ν, of the Student's t-distribution.

Constraints: $\mathtt{df} \geq 1.0$.

3.2 Optional Argument

error — type(nag_error), intent(inout), optional

The NAG *fl*90 error-handling argument. See the Essential Introduction, or the module document nag_error_handling (1.2). You are recommended to omit this argument if you are unsure how to use it. If this argument is supplied, it *must* be initialized by a call to nag_set_error before this procedure is called.

4 Error Codes

Fatal errors (error%level = 3):

error%code	Description
301	An input argument has an invalid value.

Warnings (error%level = 1):

error%code	Description
101	The solution is too close to zero to be determined accurately. Consequently, the returned value will have a poor relative precision, but the absolute value is a good approximation. This error will only occur if the degree of freedom is unity.
102	The accuracy of the result is doubtful. Convergence in the calculation of the inverse beta value was not achieved. Nevertheless, the result should be a reasonable approximation to the correct solution.

5 Examples of Usage

A complete example of the use of this procedure appears in Example 1 of this module document.

6 Further Comments

6.1 Algorithmic Detail

Given the lower tail probability p for a Student's t-distribution with ν degrees of freedom, the deviate t_p associated with p is defined as the solution to:

$$P(T < t_p : \nu) = p = \frac{\Gamma((\nu+1)/2)}{\sqrt{\nu\pi}\Gamma(\nu/2)} \int_{-\infty}^{t_p} \left(1 + \frac{T^2}{\nu}\right)^{-(\nu+1)/2} dT, \quad \nu \geq 1, \quad -\infty < t_p < \infty.$$

For $\nu = 1$ or 2 the integral equation is easily solved for t_p.
For other values of $\nu < 3$ a transformation to the beta distribution is used.
For $\nu \geq 3$ an inverse asymptotic expansion of Cornish–Fisher type is used. The algorithm is described in Hill [1].

6.2 Accuracy

The accuracy is limited to 5 significant digits for most parameter values. See [1] for discussion of the error property of various parameter values.

Example 1: Calculation of probabilities and deviate for a Student's t-distribution

This example program shows how `nag_t_prob` returns the probabilities for various parts of a Student's t-distribution given the degrees of freedom, *df*. It also shows how `nag_t_deviate` calculates the deviates (`t_calculated`) associated with the various probabilities.

1 Program Text

Note. The listing of the example program presented below is double precision. Single precision users are referred to Section 5.2 of the Essential Introduction for further information.

```
PROGRAM nag_t_dist_ex01

  ! Example Program Text for nag_t_dist
  ! NAG fl90, Release 3. NAG Copyright 1997.

  ! .. Use Statements ..
  USE nag_examples_io, ONLY : nag_std_out
  USE nag_t_dist, ONLY : nag_t_prob, nag_t_deviate
  ! .. Implicit None Statement ..
  IMPLICIT NONE
  ! .. Intrinsic Functions ..
  INTRINSIC KIND
  ! .. Parameters ..
  INTEGER, PARAMETER :: wp = KIND(1.0D0)
  ! .. Local Scalars ..
  INTEGER :: i
  REAL (wp) :: df, prob, t, t_calculated
  ! .. Local Arrays ..
  CHARACTER (1) :: tail(4)
  ! .. Executable Statements ..

  WRITE (nag_std_out,*) 'Example Program Results for nag_t_dist_ex01'

  WRITE (nag_std_out,*)
  WRITE (nag_std_out,*) 'TAIL    T      DF     PROB      DEVIATE'
  WRITE (nag_std_out,*)

  t = 0.85_wp
  df = 10.0_wp
  tail = (/ 'L', 'S', 'C', 'U'/)

  DO i = 1, 4

    prob = nag_t_prob(tail(i),t,df)
    t_calculated = nag_t_deviate(tail(i),prob,df)

    WRITE (nag_std_out,'(2X,A,4x,F6.3,F8.3,F8.4,f10.4)') tail(i), t, df, &
     prob, t_calculated
  END DO

END PROGRAM nag_t_dist_ex01
```

2 Program Data

None.

3 Program Results

```
Example Program Results for nag_t_dist_ex01

TAIL    T       DF      PROB      DEVIATE

 L     0.850  10.000  0.7924     0.8500
 S     0.850  10.000  0.4152     0.8500
 C     0.850  10.000  0.5848     0.8500
 U     0.850  10.000  0.2076     0.8500
```

Additional Examples

Not all example programs supplied with NAG *fl*90 appear in full in this module document. The following additional examples, associated with this module, are available.

`nag_t_dist_ex02`

Calculation of the deviate associated with any probability of a Students' t-distribution with known degrees of freedom.

References

[1] Hill G W (1970) Student's t-distribution *Comm. ACM* **13** 617–619